

#### 課題 4

BASIC の文法の理解が十分かどうかの確認のための読解問題である。以下のプログラムの内容を解読し付属の問いに答えよ。

##### プログラム 1 (データの並べ替え)

a,b,c を入力して大きい順に並べ替えるプログラム。sort のプログラムのと呼ばれる。

```
10 input a,b,c
20 if a=0 and b=0 and c=0 then goto 170
30 if a>=b then goto 70
40 x=a
50 a=b
60 b=x 70 if a>=c then goto 110
80 x=a
90 a=c
100 c=x
110 if b>=c then goto 150
120 x=b
130 b=c
140 c=x
150 print a,b,c
160 goto 10
170 end
```

問 小さい順に並べるにはプログラムをどのように書き直せばよいか。150行目のプリントの順を替えただけのものは認めない。書きなおしたプログラムと、なぜその様に改変したかの説明を加え、提出せよ。

## プログラム 2 (ユークリッドの互除法)

合計 5 組の 2 つ整数に対してその最大公約数と最小公倍数を求めているプログラムである。

```
10 print "a","b"," gcm","lcm"
20 for i=1 to 5
30 read l,m
40 print l,m," "; 50 c=l*m
60 if l>=m then goto 100
70 w=l
80 l=m
90 m=w
100 r=l-int(l/m)*m
110 l=m
120 m=r
130 if r>0 then goto 100
140 c=c/l
150 print l,c
160 next i
170 data 10,20,5,25,588,133
180 data 58,369,1230,105
```

問 3 個の整数の最大公約数と最小公倍数を求めるにはどのようにプログラムを書き直せばよいか。書きなおしたプログラムと、なぜその様に改変したかの説明を加え、提出せよ。

**プログラム 3** (ニュートン法による根の計算)

$b$  の  $n$  乗根をニュートン法を用いて求めるプログラムである。ニュートン法については微分積分の教科書にはたいてい載っているから、各自確認しておくこと。

```
10 input "n,b";n,b
20 x=1
30 f=x^n-b
40 d1=n*x^(n-1)
50 x1=x-f/d1
60 e=1*10^(-6)
70 if abs(f)<e then goto 100
80 x=x1
85 print x
90 goto 30
100 end
```

問 与えられた  $b$  に対して  $\frac{x}{x^2+1} = b$  となる  $x$  を求めるには、このプログラムをどのように書き直せばよいか。書き直したプログラムと、なぜそのように書き直したかの説明を加えて提出せよ。

プログラム 4 (定積分の台形公式による近似値の計算)

定積分  $\int_0^1 \frac{dx}{1+x}$  の近似値を台形公式を用いて求めるプログラムである。  
台形公式については微積の教科書に載っているから各自確認しておくこと。

```
0 rem daikei formula
10 def fnf(x)=1/(1+x)
20 a=0
30 b=1
40 n=10
50 h=(b-a)/n
60 s=0
70 x=a+h
80 j=n-1
90 for i=1 to j
100 s=s+fnf(x)
110 x=x+h
120 next i
130 s=h*(fnf(a)+fnf(b)+2*s)/2
140 print s
150 end
```

問 a, b, n をいくつか変化させ真の値と比較し考察せよ。その考察をまとめて提出せよ。

プログラム 5 (連立方程式の解法 (掃出し法))

3元の連立方程式を掃出し法により解くプログラムである。掃出し法についてはたいていの線形代数の教科書に載っているから、各自確認しておくこと。

```
10 dim a(10,11)
20 m=3
30 n=m+1
40 for i=1 to m
50 for j=1 to n
60 read d
70 a(i,j)=d
80 next j
90 next i
100 for k=1 to m
110 p=a(k,k)
120 for j=k to n
130 a(k,j)=a(k,j)/p
140 next j
150 for i=1 to m
160 if (i-k)=0 then goto 210
170 a1=a(i,k)
180 for j=k to n
190 a(i,j)=a(i,j)-a1*a(k,j)
200 next j
210 next i
220 next k
230 for i=1 to m
240 print "x(";i;")=",a(i,n)
250 next i
260 end
270 data 2,1,-3,1,4,1,-2,4,3,2,-1,6
```

問 このプログラムの意味を行番号に沿って説明せよ。つぎに4元の連立方程式を実際に解いてみよ。またこのプログラムで計算できない連立方程式の例を与え、その理由を述べよ。